Algorithm AS 275: Computing the Non-Central χ^2 Distribution Function
Author(s): Cherng G. Ding
Source: *Journal of the Royal Statistical Society. Series C (Applied Statistics)*, Vol. 41, No. 2 (1992), pp. 478-482
Published by: Wiley for the Royal Statistical Society
Stable URL: http://www.jstor.org/stable/2347584
Accessed: 28/04/2014 15:00

```
      HII = ZERO
      DO 20 COL = 1, NREQ
         IF (SQRT(D(COL)) .LE. TOL(COL)) THEN
            WK(COL) = ZERO
            GO TO 20
         END IF
         POS = COL - 1
         SUM = XROW(COL)
         DO 10 ROW = 1, COL-1
            SUM = SUM - WK(ROW) * RBAR(POS)
            POS = POS + NP - ROW - 1
   10    CONTINUE
         WK(COL) = SUM
         HII = HII + SUM**2 / D(COL)
   20 CONTINUE
C
      RETURN
      END
```

Algorithm AS 275

Computing the Non-central χ^2 Distribution Function

By Cherng G. Ding†

National Chiao Tung University, Taipei, Republic of China

[Received September 1990]

Keyword: Series representation

Language

Fortran 77

Description and Purpose

The algorithm presented in this paper for computing the non-central χ^2 distribution function is based on an alternative series representation, which is an infinite weighted sum of central χ^2-densities. A bound has been obtained for the error incurred by truncating this series. The accuracy of recursive computation for the non-central χ^2 distribution function can be controlled with this bound.

Notation and Method

The non-central χ^2 distribution function with positive degrees of freedom f and non-negative non-centrality parameter θ may be expressed as a series of central χ^2 distribution functions with Poisson weights. More precisely (see, for example, Johnson and Kotz (1970)):

†*Address for correspondence*: Institute of Management Science, National Chiao Tung University, 4F, 114 Chung-Hsiao W. Road, Section 1, Taipei, Taiwan, 10012, Republic of China.

$$\Pr\{\chi^2(f,\theta) \leqslant x\} = \sum_{i=0}^{\infty} \frac{\exp(-\lambda)\lambda^i}{i!} \Pr\{\chi^2(f+2i) \leqslant x\}, \tag{1}$$

where $\lambda=\theta/2$. Farebrother (1987) presented an algorithm to approximate the infinite series (1) for non-negative integer degrees of freedom by the finite sum of the first K terms. The value of K was chosen so that the upper bound on the truncation error was less than some predetermined small number. An auxiliary algorithm for the standard normal distribution is necessary. Posten (1989) provided a recursive algorithm for evaluating the non-central χ^2 distribution function with positive real degrees of freedom in terms of a single central χ^2 distribution function. In this paper, another series representation for $\Pr\{\chi^2(f,\theta) \leqslant x\}$ is derived, and then a bound for the error incurred by truncating this series is obtained. A simple recursive algorithm is given without any auxiliary normal or central χ^2-routine. It applies for positive real degrees of freedom.

It is easy, by using integration by parts, to show that

$$\Pr\{x^2(f) \leqslant x\} = 2f(x;f+2) + \Pr\{x^2(f+2) \leqslant x\}, \tag{2}$$

where $f(x;r)$ denotes the central χ^2-density with r degrees of freedom. Equation (2) also holds for the non-central case (see equation (12) in Cohen (1988)). The series representation for the central χ^2 distribution function follows:

$$\Pr\{x^2(f+2i) \leqslant x\} = \sum_{k=i}^{\infty} 2f(x;f+2+2k) = \sum_{k=i}^{\infty} t_k, \qquad i=0,1,2,\ldots, \tag{3}$$

where the terms t_k are evaluated by using the recursion

$$t_0 = 2f(x;f+2) = \frac{1}{\Gamma(f/2+1)}\left(\frac{x}{2}\right)^{f/2} \exp\left(-\frac{x}{2}\right),$$

$$t_k = t_{k-1}\frac{x}{f+2k}, \qquad k=1,2,3,\ldots. \tag{4}$$

Substituting equation (3) into equation (1) and then doing some simple algebraic operations, we have a new series representation for the non-central χ^2-distribution as follows, which is an infinite weighted sum of central χ^2-densities.

$$\begin{aligned}\Pr\{\chi^2(f,\theta) \leqslant x\} &= \sum_{i=0}^{\infty} 2\exp(-\lambda)\left(\sum_{k=0}^{i}\frac{\lambda^k}{k!}\right) f(x;f+2+2i)\\ &= \sum_{i=0}^{\infty}\left(\sum_{k=0}^{i} u_k\right) t_i\\ &= \sum_{i=0}^{\infty} v_i t_i, \end{aligned} \tag{5}$$

where $\lambda=\theta/2$, and

$$v_0 = u_0 = \exp(-\lambda),$$

$$v_i = v_{i-1} + u_i, \qquad u_i = u_{i-1}\lambda/i, \quad i \geqslant 1,$$

$$t_i, \quad i \geqslant 0, \qquad \text{as in equation (4).}$$

Considering equation (3) for $i=0$, we have

$$\Pr\{\chi^2(f) \leqslant x\} = \sum_{k=0}^{\infty} t_k = \sum_{k=0}^{n-1} t_k + \Pr\{x^2(f+2n) \leqslant x\}, \tag{6}$$

where $\Sigma_{k=0}^{n-1} t_k$ is the truncated series and

$$\Pr\{\chi^2(f+2n) \leqslant x\} = \sum_{k=n}^{\infty} t_k,$$

the truncation error. It can be shown, by the mean value theorem, that $\Pr\{\chi^2(f+2n) \leqslant x\} \leqslant t_{n-1}x/(f+2n-x)$ if $f+2n > x$. Let E_n be the truncation error at $i=n$ for series (5). Since $v_i \leqslant 1$ for all $i \geqslant 0$, it follows that if $f+2n > x$

$$E_n = \sum_{i=n}^{\infty} v_i t_i \leqslant \sum_{i=n}^{\infty} t_i \leqslant t_{n-1}x/(f+2n-x). \tag{7}$$

Recursive iterations for computing series (5) are performed until the error bound above is less than a predetermined small number (see ERRMAX in the section on constants). This error bound is a common criterion for computing equations (5) and (6).

Structure

REAL FUNCTION CHI2NC(X, F, THETA, IFAULT)

Formal parameters

X	Real	input:	the percentage point x (at which the cumulative probability is desired)
F	Real	input:	the degrees of freedom f
THETA	Real	input:	the non-centrality parameter θ
IFAULT	Integer	output:	a fault indicator: =1 if the desired accuracy could not be obtained within ITRMAX iterations (see the sections on constants and time and accuracy); =2 if $f \leqslant 0$ or $\theta < 0$; =3 if $x < 0$; =0 otherwise

Auxiliary Algorithms

Function ALOGAM (Pike and Hill, 1966) is used to compute the natural logarithm of the gamma function; alternatively function ALNGAM (Macleod, 1989) could be used.

Constants

The variables ERRMAX and ITRMAX are set by the DATA statement in CHI2NC. ERRMAX denotes a bound on the truncation error. The value given here is 1.0×10^{-6}. ITRMAX is an integer to control the number of iterations. The value given here is 50.

Time and Accuracy

The execution time is a non-decreasing function of x. If x is fixed, then the error bound in inequality (7) decreases, but the execution time increases as the number of iterations n increases. A compromise between time and accuracy can be made with ERRMAX and ITRMAX defined in the section on constants. The values of these variables may be altered to suit the user's needs. Moreover, if the series does not converge within ITRMAX (e.g. 50) iterations for a predetermined error bound ERRMAX (e.g. 1.0×10^{-6}), the value of ITRMAX may be enlarged to obtain the result with the accuracy required.

Precision

Double-precision operation may be performed by making the following changes.

(a) Change REAL to DOUBLE PRECISION in the FUNCTION statement and in the REAL statement.
(b) Change the constants in the DATA statements to double precision.
(c) Change EXP to DEXP and INT to IDNINT.
(d) Change the value of ERRMAX from 1.0×10^{-6} to 1.0×10^{-12} and the value of ITRMAX from 50 to 100.
(e) Make appropriate modifications to the auxiliary routine ALOGAM or ALNGAM.

Acknowledgement

This work was partially supported by the National Science Council, Republic of China.

References

Cohen, J. D. (1988) Noncentral chi-squared: some observations on recurrence. *Am. Statistn* **42**, 120–122.

Farebrother, R. W. (1987) Algorithm AS 231: The distribution of a noncentral χ^2 variable with nonnegative degrees of freedom. *Appl. Statist.*, **36**, 402–405.

Johnson, N. L. and Kotz, S. (1970) *Distributions in Statistics: Continuous Univariate Distributions- in*, p. 132. New York: Wiley.

Macleod, A. J. (1989) Algorithm AS 245: A robust and reliable algorithm for the logarithm of the gamma function. *Appl. Statist.*, **38**, 397–402.

Pike, M. C. and Hill, I. D. (1966) Algorithm 291: Logarithm of the gamma function. *Communs As Comput. Mach.*, **9**, 684.

Posten, H. O. (1989) An effective algorithm for the noncentral chi-squared distribution function. *Am. Statistn*, **43**, 261–263.

```
      REAL FUNCTION CHI2NC(X, F, THETA, IFAULT)
C
C        ALGORITHM AS 275 APPL.STATIST. (1992), VOL.41, NO.2
C
C        Computes the noncentral chi-squared distribution function
C        with positive real degrees of freedom f and nonnegative
C        noncentrality parameter theta
C
      REAL X, F, THETA
      INTEGER IFAULT
C
      INTEGER ITRMAX
      LOGICAL FLAG
      REAL ERRMAX, ZERO, ONE, TWO, LAM, N, U, V, X2, F2, T, TERM,
     *     BOUND, ALOGAM
C
      EXTERNAL ALOGAM
C
      DATA ERRMAX, ITRMAX / 1.0E-6, 50 /
      DATA ZERO, ONE, TWO / 0.0, 1.0, 2.0 /
C
      CHI2NC = X
      IFAULT = 2
      IF (F .LE. ZERO .OR. THETA .LT. ZERO) RETURN
      IFAULT = 3
      IF (X .LT. ZERO) RETURN
      IFAULT = 0
      IF (X .EQ. ZERO) RETURN
      LAM = THETA / TWO
C
C        Evaluate the first term
C
      N = ONE
      U = EXP(-LAM)
      V = U
      X2 = X / TWO
      F2 = F / TWO
      T = X2 ** F2 * EXP(-X2) / EXP(ALOGAM((F2 + ONE), IFAULT))
C
C        There is no need to test IFAULT since the value of F has
C        already been checked
C
      TERM = V * T
      CHI2NC = TERM
C
C        Check if (f+2n) is greater than x
C
      FLAG = .FALSE.
   10 IF ((F + TWO * N - X) .LE. ZERO) GO TO 30
C
C        Find the error bound and check for convergence
C
      FLAG = .TRUE.
   20 BOUND = T * X / (F + TWO * N - X)
      IF (BOUND .GT. ERRMAX .AND. INT(N) .LE. ITRMAX) GO TO 30
      IF (BOUND .GT. ERRMAX) IFAULT = 1
      RETURN
C
C        Evaluate the next term of the expansion and then the
C        partial sum
C
   30 U = U * LAM / N
      V = V + U
      T = T * X / (F + TWO * N)
      TERM = V * T
      CHI2NC = CHI2NC + TERM

      N = N + ONE
      IF (FLAG) GO TO 20
      GO TO 10
C
      END
```